10

VECTOR FIELDS

10.1. Vector fields

▼ VectorPlot[ ]

★ **VectorPlot**$[\{v_x, v_y\}, \{x, x_{min}, x_{max}\}, \{y, y_{min}, y_{max}\}]$

It plots the vector field {y , x} in each point of the plane.

```
Clear["Global`*"]
vecf1 = VectorPlot[{y, x}, {x, -3, 3}, {y, -3, 3},
  VectorStyle → RGBColor[0.5, 0, 0.7], VectorScale → Small]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
VectorPlot[{{-x, y}, {y, x}}, {x, -3, 3}, {y, -3, 3}]
```



▼ Some options of the command VectorPlot[ ]

★ VectorColor and VectorScale

```
vecf2 = VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1},
  VectorScale → Medium, VectorColorFunction → "Rainbow"]
```



★ Stream

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, StreamPoints → 15, StreamStyle → Purple,
 StreamScale → Full, VectorPoints → 12, VectorScale → Medium, VectorStyle → Automatic]
```



★ Fields

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, RegionFunction → Function[{x, y}, x y < 0]]
```



★ **Vector fields in 3D**

```
VectorPlot3D[{x, y, z}, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```



10.2. Solving Differential Equations

▼ DSolve[equation, function, variable]

It finds the solution y(x) of the Differential Equation

★ **Solution of the First order Differential Equation**

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4\,y[x] + y'[x] == 0$

```
S1 = DSolve[ed1, y[x], x]
```

$\{\{y[x] \to e^{-4x}\,C[1]\}\}$

★ **Solution of the Second order Differential Equation**

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2\,y[x] - 3\,y'[x] + y''[x] == 0$

```
S2 = DSolve[ed2, y[x], x]
```

$\{\{y[x] \to e^{x}\,C[1] + e^{2x}\,C[2]\}\}$

★ **NDsolve[equation, function, {x, x_{min}, x_{max}}]**

It calculates the numerical solution of a Differential Equation

```
s = NDSolve[{y'[x] == y[x] Cos[x + y[x]], y[0] == 1}, y, {x, 0, 30}]
```

```
{{y → InterpolatingFunction[{{0., 30.}}, <>]}}
```


E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
```



10.3. First Order Differential Equations

▼ General solution and particular solution

★ General solution of a Differential Equation

Based in a solution that we have obtained, the solution function of the Differential Equation is defined. This solution will depend on a parameter

```
so[x_, c_] = S1[[1, 1, 2]] /. C[1] → c
```

$c\, e^{-4x}$

★ Family of solutions

Giving values to the parameter “c”, a family of solutions is built

```
listsol = Table[so[x, c], {c, -4, 4, 1}]
```

$\{-4 e^{-4x}, -3 e^{-4x}, -2 e^{-4x}, -e^{-4x}, 0, e^{-4x}, 2 e^{-4x}, 3 e^{-4x}, 4 e^{-4x}\}$

```
famsol = Plot[Evaluate[listsol], {x, 0, 1},
  PlotStyle → Thickness[0.005], PlotRange → {-4, 4}]
```



★ Differential Equation with initual value

The steps that we have to follow to solve the following Differential Equation with initial value y(0)=2 are:

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4 y[x] + y'[x] == 0$

```
solo1 = DSolve[{ed1, y[x0] == y0}, y[x], x]
```

$\{\{y[x] \to e^{-4x+4x0}\, y0\}\}$

```
yg[x_] = solo1[[1, 1, 2]]
```

$e^{-4x+4x0}\, y0$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
yg[x] /. {x0 → 0, y0 → 2}
```

$2 e^{-4 x}$

```
sp = DSolve[{ed1, y[0] == 2}, y[x], x]
```

$\{\{y[x] \to 2 e^{-4 x}\}\}$

```
yp[x_] = sp[[1, 1, 2]]
```

$2 e^{-4 x}$

The steps that we have to follow to solve the following Differential Equation with initial values y(0)=2 and y'(0)=1 are:

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2 y[x] - 3 y'[x] + y''[x] == 0$

```
sp2 = DSolve[{ed2, y[0] == 2, y'[0] == 1}, y[x], x]
```

$\{\{y[x] \to -e^x (-3 + e^x)\}\}$

```
yp2[x_] = sp2[[1, 1, 2]]
```

$-e^x (-3 + e^x)$

▼ Some remarks

Example: Given the Differential Equation x^2*y'[x]+y[x]^2==0 and based in a general solution, we can find a family of solutions:

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$\left\{\left\{y[x] \to -\frac{x}{1 + x C[1]}\right\}\right\}$

```
solor1[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$-\frac{x}{1 + c x}$

```
listsol = Table[solor1[c, x], {c, -2, 2, .5}]
```

$\left\{-\frac{x}{1 - 2. x}, -\frac{x}{1 - 1.5 x}, -\frac{x}{1 - 1. x}, -\frac{x}{1 - 0.5 x}, -1. x, -\frac{x}{1 + 0.5 x}, -\frac{x}{1 + 1. x}, -\frac{x}{1 + 1.5 x}, -\frac{x}{1 + 2. x}\right\}$

```
familysol = Plot[Evaluate[listsol], {x, -2, 3}]
```



If we try to obtain the solution of the initial value Differential Equation {x^2*y'[x]+y[x]^2==0,y[0]==2} using DSolve, we do not obtain any solution:

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 2}, y[x], x]
```

DSolve::bvnul : For some branches of the general solution, the given boundary conditions lead to an empty solution. ≫

{}

If we try to solve it by obtaining a general solution of the Differential Equation and after that calculating the value of "c" in the way in which the initial values are verified, we do not obtain solution either. It can be observed in the graph that the solutions of the Differential Equation do not pass from the point (0,2).

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\, C[1]}\right\}\right\}$$

```
solor[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$$-\frac{x}{1 + c\, x}$$

```
kons = Solve[solor[c, 0] == 2, c]
```

{}

If we try to obtain the solution of the initial value Differential Equation {x^2*y'[x]+y[x]^2==0,y[0]==0} using DSolve, we obtain infinite solutions. It can be observed in the graph that there are many solutions of the Differential Equation that pass from the point (0,2).

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 0}, y[x], x]
```

DSolve::bvnr : For some branches of the general solution, the given boundary conditions do not restrict the existing freedom in the general solution. ≫

DSolve::bvsing : Unable to resolve some of the arbitrary constants in the general solution using the given boundary conditions. It is possible that some of the conditions have been specified at a singular point for the equation. ≫

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\, C[1]}\right\}\right\}$$

10.4. Vector field joined to a Differential Equation

▼ Field of the tangents

Given the Differential Equation y'=f(x,y), the vector field joined to this Differential Equation is given by {1,f(x,y)}

★ **Given the Differential Equation y'=x, we will calculate the general solution and we will plot the family of solutions**

```
ed = y'[x] == x
```

y′[x] == x

```
S = DSolve[ed, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{x^2}{2} + C[1]\right\}\right\}$$

```
so[x_, c_] = S[[1, 1, 2]] /. C[1] → c
```

$$c + \frac{x^2}{2}$$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
listsol = Table[so[x, c], {c, -5, 8, 1}]
```

$$\left\{-5+\frac{x^2}{2},\ -4+\frac{x^2}{2},\ -3+\frac{x^2}{2},\ -2+\frac{x^2}{2},\ -1+\frac{x^2}{2},\ \frac{x^2}{2},\ 1+\frac{x^2}{2},\ 2+\frac{x^2}{2},\ 3+\frac{x^2}{2},\ 4+\frac{x^2}{2},\ 5+\frac{x^2}{2},\ 6+\frac{x^2}{2},\ 7+\frac{x^2}{2},\ 8+\frac{x^2}{2}\right\}$$

```
famsol = Plot[Evaluate[listsol], {x, -5, 5},
  PlotStyle → {Green, Thickness[0.005]}, PlotRange → {-5, 8}]
```



★ **VectorPlot**[{v_x, v_y}, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}]

Given the Differential Equation y'=x, the field of vector tangents is given by {1,x}

```
vecfield = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorStyle → RGBColor[0.2, 0.8, 0.3], VectorScale → Small]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

★ **Joining all the previous graphs**

```
Show[{vecfield, famsol}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ **The option "stream"**

```
vecfield2 = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, StreamScale → Full, StreamPoints → 15, StreamScale → Full]
```



▼ Orthogonal trajectories

Given the Differential Equation y′ = x, the differential equation that corresponds to the orthogonal trajectories of its solution curves is given by y′ = -1/x, being the tangent vector m′= {- f (x, y).1} = {- x,1}

★ **The Differential Equation of the orthogonal trajectories is solved, the general solution is defined and the family of curves is plotted**

```
edto = y'[x] == -1 / x
```

$$y'[x] == -\frac{1}{x}$$

```
Sto = DSolve[edto, y[x], x]
{{y[x] → C[1] - Log[x]}}
sgto[x_, c_] = Sto[[1, 1, 2]] /. C[1] → c
c - Log[x]
listsol1 = Table[sgto[x, c], {c, -5, 8, 1}]
{-5 - Log[x], -4 - Log[x], -3 - Log[x], -2 - Log[x], -1 - Log[x], -Log[x], 1 - Log[x],
 2 - Log[x], 3 - Log[x], 4 - Log[x], 5 - Log[x], 6 - Log[x], 7 - Log[x], 8 - Log[x]}
listsol2 = Table[sgto[-x, c], {c, -5, 8, 1}]
{-5 - Log[-x], -4 - Log[-x], -3 - Log[-x], -2 - Log[-x], -1 - Log[-x], -Log[-x], 1 - Log[-x],
 2 - Log[-x], 3 - Log[-x], 4 - Log[-x], 5 - Log[-x], 6 - Log[-x], 7 - Log[-x], 8 - Log[-x]}
famsol1 = Plot[Evaluate[listsol1], {x, 0.01, 3},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
famsol2 = Plot[Evaluate[listsol2], {x, -3, -0.01},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
famsolto = Show[{famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



If the tangent vector of the vector field joined to the Differential Equation y'=x is given by {1,f(x,y)}={1,x}, then the tangent vector of the orthogonal trajectories is {-x,1}

```
g1 = VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, VectorPoints → 12, VectorStyle → {Purple, Blue}]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
g2 = Show[{famsol, famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ **Family of curves and field of the tangents in the same figure**

```
Show[{g1, g2}, PlotRange → {{-3, 3}, {-3, 8}}]
```




```
VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8}, StreamPoints → 15,
 StreamScale → Full, VectorPoints → 12, VectorScale → Small, VectorStyle → {Purple, Blue}]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez